

Megjegyzések. 1. Mindkét megoldás teszteli a beütött adatokat, és csak a 10-nél nem nagyobb természetes számokat fogadja el. Feliratokkal segíti a kezelőt az adatok beadásakor és a végeredmény közlésekor.

2. A két program hossza nem tér el lényegesen. Kevés adat esetében sem a futásidőben, sem az adattároláshoz szükséges memóriakapacitásban nincs nagy eltérés. De gondoljuk meg, melyiket tudjuk használni, ha pl. 10 000 adatot kell feldolgozni (az adatokat egy automata mérőkészülék adja...)!

```
5  N=20 : K=10
10  DIM L (N)
20  CLS : PRINT "LOLAP KIERTEKELES"
30  FOR I=1 TO N
40  PRINT I; ". LOVES EREDMENYE"; : INPUT X
50  IF X<0 OR X>10 OR X>INT X THEN
    PRINT "TEVES ADAT" : GOTO 40
60  L (I)=X
70  NEXT I
80  P=0
90  PRINT : PRINT : PRINT : PRINT "A LEGJOBB ";K;" LOVES :"
100 FOR I=1 TO K
110 FOR J=I+1 TO N
120 IF L(J)>L(I) THEN X=L(I) : L(I)=L(J) : L(J)=X
130 NEXT J
140 PRINT I,L(I)
150 P=P+L(I)
160 NEXT I
170 PRINT : PRINT "OSSZPONTSZAM =";P : PRINT "HA EL
    OLVASTAD, NYOMJ LE EGY GOMBOT"
180 A$=INKEY $
190 IF INKEY $="" THEN 190
200 RUN
```

3. Eljárásaink az *A*, illetve *L* tömböt változatlanul megőrizték. Így egy esetleges programfejlesztés, bővítés (pl. több versenyző lőlapjainak statisztikai kiértékelése vagy a verseny végeredményének megállapítása) tovább használhatja azokat.

4. A 180–200 sorok nélkül is működik programunk. Minden egyes versenyző adatainak beírása után számol a gép, kijelzi a végeredményt, majd megáll. A következő versenyző adatainak beírásához újra el kell indítani egy RUN paranccsal. Az említett három sor begépelése esetén egyetlen tetszőleges billentyű lenyomására újra indul a program. Kísérletezzünk: elhagyható-e a 180 jelű sor?

5. Mindkét megoldás HT–1080Z Scoolcomputerre készült. Az elterjedt ZX–81 gépre is átírhatók programjaink, tulajdonképpen csak az elhagyott LET kulcsszavak leírására és több utasítást tartalmazó programsorok szétírására van szükség. A 180-200 programsorokat

```
180 PAUSE 4E4
190 RUN
```

helyettesítheti.

```
5  N=20 : K=10
10  DIM A (K)
20  CLS : PRINT "LOLAP KIERTEKELES"
30  FOR I=1 TO N
40  PRINT I; ". LOVES EREDMENYE"; : INPUT X
50  IF X<0 OR X>10 OR X>INT X THEN
    PRINT "ILYEN LOVES NINCS" : GOTO 40
60  IF X>0 THEN A(X)=A(X)+1
70  NEXT I
80  P=0 : I=K : FOR J=K TO 1 STEP -1
90  IF A(J)>I THEN A(J)=I
100 P=P+J*A(J); I = I - A(J) : NEXT J
110 PRINT "A LOVESZ ALTAL ELERT EREDMENYT : ";P;
    " KOR"
180 A$=INKEY $
190 IF INKEY $="" THEN 190
200 RUN
```